计算机应用与教学系列

工程力学的计算机解法

夏永源　任　芬　编著

科学出版社

工程力学的计算机解法

科

计算机应用与教学系列

工程力学的计算机解法

夏永源　任　芬　编著

科　学　出　版　社

1994

(京)新登字 092 号

内容简介

本书系统地介绍了工程力学问题的计算机解法，内容包括静力学、运动学和动力学三个部分. 本书将力学模型、数学模型、计算方法和计算程序有机地结合成为一个整体，使工程力学中原来计算繁难的问题变得十分容易求解. 由于计算程序都是通用程序，适用范围广泛.

本书多数程序采用 BASICA 语言写成，少数程序采用 True BASIC 语言写成，可在 IBM-PC 机或其兼容机上运行. 全部源程序、例题、习题和答案已存储于三张 5.25 英寸的软磁盘上，便于读者上机算题. 与本书配套的软件将与本书同时出版发行。

本书适用于高等理工科院校的广大师生，也可供有关专业的工程技术人员参考.

计算机应用与教学系列

工程力学的计算机解法

夏永源　任　芬　编著

责任编辑　张建荣

科学出版社　出版

北京东黄城根北街 16 号

邮政编码：100717

北京市桦星电脑技术应用部激光照排

天津市蓟县燎原印刷厂印刷

新华书店北京发行所发行　各地新华书店经售

*

1994 年 1 月第　一　版　开本：787×1092 1/16

1994 年 1 月第一次印刷　印张：17 ¼

印数：1—1700　字数：400000

ISBN 7-03-003758-8/TP・286

定价：20.00 元

前　言

任何学习过工程力学这门课的人，都有这种印象：理论部分不难理解，但工程力学的问题灵活多变，某些问题计算繁复，不易对付. 长期以来，人们一直在寻找一种统一的、按一定步骤进行顺序计算的方法，以解决各种各样的工程力学问题.

在理论方面，法国著名数学家和力学家拉格朗日提出了解决上述问题的方法. 他在《分析力学》一书中写道："我在这里所提出的方法，既不需要几何性质的也不需要力学性质的作图或推论，而只要求按照一种正规和一致的方案进行代数运算. 那些爱好分析的人会愉快地看到力学变成了数学的一个分支，并且会感谢我这样扩大了分析的领域".

在应用问题的解决方面，解题人在按照一种正规的和统一的步骤作了力学性质的和几何性质的分析和作图之后，能否编一个计算机程序，该程序将自动生成方程（如静力学中的平衡方程和和动力学中的运动微分方程），并对方程进行求解、打印数值结果、以曲线形式显示数值结果. 把大量的、费神的运算工作交给电子计算机去完成，也一直是许多力学工作者的愿望.

文献[1]的作者们曾于 1985 年作了有益的尝试，几年来的教学实践证明，很受师生欢迎. 文献[3]的作者又对这些程序作了改进. 本书作者除使原有 4 个程序更符合实用要求外，在程序数量上已增为 12 个，内容大大地扩充了. 补充了更多的算例，增设了习题和答案. 计算结果除给出数值外，更注重图示，增强直观性. 新增部分都是动力学方面的，但是没有动量定理、动量矩定理和动能定理等方面的内容，因为这些问题比较简单、运算少，不需要计算机来完成. 而且，它们事实上是质点运动微分方程和拉格朗日方程的等价形式或首次积分.

过去，在依靠手算的时代，复杂的动力学问题的研究由于运动微分方程组积分求解析解方面的困难，常常被迫半途而废. 如今，有了 Runge-Kutta 法这样一些自动开始的、精度又高的计算方法，我们已能把许多复杂的动力学问题的研究进行到底，直接获得数值形式的运动规律和图形. 读者将会发现，本书各程序已覆盖了工程力学的最主要部分. 任何只要粗通工程力学和 BASIC 语言的读者，都能很快掌握本书内容，走上自己独立算题的道路，进入一新的环境.

对于爱好编写程序的读者来说，本书各程序在计算方法、编程技巧和输入输出格式等方面的经验，都是值得借鉴和引用的.

作者将多年来的科研成果在本书的算例、源程序、习题中介绍出来，其内容是航空、航天、建筑、动力、机械等广泛的工程领域普遍感到兴趣的.

全部源程序、例题、习题和答案都记录在三张 5.25 英寸的软盘上，随时可以运行. 12 个源程序中有 10 个用 BASICA 语言编写，2 个用 True BASIC 语言编写. 所用计算机为 IBM-PC 或其兼容机.

本书可供高等理工科院校广大师生阅读，也可供工程技术人员参考.

限于作者水平，疏漏之处在所难免，切望读者批评、指正.

夏永源　任　芬

1992 年 12 月

目　　录

第一章　静力学(STATICS. BAS)

本程序可以用来求解平面任意力系、平面桁架、空间任意力系、空间桁架和摩擦的各种平衡问题.算题人只需按人机对话方式,根据屏幕提问,逐步输入载荷数据和几何参数数据,程序便可自动形成平衡方程组,并求得全部未知外力和内力.随着算题结束,程序已自动形成一个新的数据文件,它包括了全部输入数据和输出数据,供打印之用.

```
LIST
100 REM COMPUTER-AIDED INSTRUCTIONS OF ENGINEERING MECHANICS.
110 REM PROGRAM NAME:--------------------------------------------------------------------------------STATICS. BAS
120 REM CONTENTS: (1) PLANAR FORCE SYSTEMS.
130 REM           (2) PLANAR TRUSSES.
140 REM           (3) SPACIAL FORCE SYSTEMS.
150 REM           (4) SPACIAL TRUSSES.
160 REM           (5) SLIDING FRICTIONS.
170 DIM X(30),Y(30),Z(30),A(90,91),G(90)
180 OPEN "B:TEMPORAL. DAT" FOR OUTPUT AS #1 'EXAMPLE:EXAMPL9. DAT,PROBLM3. DAT,...
190 PRINT "ENTER YOUR LEARNING NUMBER.......................................NUM =";
200 INPUT NUM
210 PRINT "ENTER FILE NAME..................................................FILENAME =";
220 INPUT CC$
230 PRINT #1,"LEARNING NUMBER..........................................................NUM =";NUM
240 PRINT #1,"FILE NAME........................................................FILENAME =";CC$
250 PRINT #1,
260 GOTO 290
270 A$="A":INPUT " * * * DO YOU HAVE NEXT PROBLEM? (Y/N)",A$
280 IF A$="n" OR A$="N" THEN 660
290 PRINT " * * * * * * * * * * * * * * * * * INPUT DATA * * * * * * * * * * * * * * * * *"
300 PRINT #1," * * * * * * * * * * * * * * * * INPUT DATA * * * * * * * * * * * * * * * * * * *"
310 PRINT " * * * WHICH SUBROUTINE DO YOU WANT TO USE?"
320 PRINT "ENTER:1,     FOR PLANAR FORCE SYSTEMS."
330 PRINT "      2,     FOR PLANAR TRUSSES."
340 PRINT "      3,     FOR SPACIAL FORCE SYSTEMS."
350 PRINT "      4,     FOR SPACIAL TRUSSES."
360 PRINT "      5,     FOR SLINDING FRICTIONS."
370 PRINT "      0,     FOR ENDING YOUR WORK."
380 PRINT "   (AFTER ENTERING THE NUMBER YOU WANT,"
390 PRINT "    YOU SHOULD PRESS <ENTER> KEY.    )"
400 N6=10:EP=.0000001
410 N7=3*N6
420 GOTO 440
430 PRINT "A WRONG KEY WAS PRESSED.    DO IT AGAIN PLEASE."
440 INPUT N5
450 PRINT #1,N5
460 FOR I=1 TO N7
```

```
470 G(I)=0
480 FOR J=1 TO N7+1
490 A(I,J)=0
500 NEXT J
510 NEXT I
520 C$ = "THE FOLLOWING DATA ARE THE ANSWERS:"
530 IF N5=0 THEN 660
540 ON N5 GOTO 600,560,620,580,640
550 GOTO 430
560 GOSUB 800
570 GOTO 270
580 GOSUB 1620
590 GOTO 270
600 GOSUB 2450
610 GOTO 270
620 GOSUB 3220
630 GOTO 270
640 GOSUB 4650
650 GOTO 270
660 END
670 INPUT " * * * HOW MANY JOINTS IN TOTAL ";N1
680 PRINT #1,N1
690 INPUT " * * * HOW MANY BARS IN TOTAL ? ";N2
700 WRITE #1,N2
710 INPUT " * * * HOW MANY SUPPORTING FORCES IN TOTAL";N3
720 WRITE #1,N3
730 RETURN
740 INPUT " * * * HOW MANY UNKOWN-FORCES AND UNKNOWN-MOMENTS IN TOTAL";N4
750 WRITE #1,N4
760 INPUT " * * * HOW MANY RIGID BODIES";N2
770 WRITE #1,N2
780 RETURN
790 REM -----------------------------------------------------------------------------PLANAR TRUSSES
800 REM
810 PRINT "THIS IS THE SUBROUTINE OF PLANAR TRUSSES FROM NOW ON."
820 GOSUB 670
830 N=2*N1
840 NC=N+1
850 N4=N2+N3
860 IF N <> N4 THEN 1580
870 PRINT "ENTER EACH JOINT COORDINATE:  (X,Y)          "
880 FOR I=1 TO N1
890 PRINT "JOINT";I
900 INPUT "                         ";X(I),Y(I)
910 WRITE #1,X(I),Y(I)
920 NEXT I
930 PRINT "ENTER TWO JOINT NUMBERS FOR EACH BAR :"
940 PRINT "TRUSS NO. ,        JOINT NO. ,      (A,B)        "
950 FOR I=1 TO N2
960 PRINT "      ";I
970 INPUT "                         ";J,K
```

```
980 WRITE #1,J,K
990 X1=X(K)-X(J)
1000 Y1=Y(K)-Y(J)
1010 D=SQR(X1*X1+Y1*Y1)
1020 C=X1/D
1030 S=Y1/D
1040 L=2*J-1
1050 A(L,I)=C
1060 A(L+1,I)=S
1070 L=2*K-1
1080 A(L,I)=-C
1090 A(L+1,I)=-S
1100 NEXT I
1110 PRINT "ENTER ACTING JOINT NO. AND ANGLE(X) OF EACH SUPPORTING FORCE:"
1120 PRINT "SUPPORTING FORCE NO. ,   JOINT NO. ,   ANGLE(X)   "
1130 FOR I=1 TO N3
1140 PRINT "                    ";I
1150 INPUT "                              ";J,U
1160 WRITE #1,J,U
1170 U=U*ATN(1)/45
1180 L=2*J-1
1190 M=N2+I
1200 A(L,M)=COS(U)
1210 A(L+1,M)=SIN(U)
1220 NEXT I
1230 INPUT "*** HOW MANY KNOWN-FORCES ARE THERE ";M1
1240 WRITE #1,M1
1250 IF M1 <=0 THEN 1360
1260 PRINT "      JOINT NO. ,    MAGNITUDE,     ANGLE(X) "
1270 FOR L1=1 TO M1
1280 PRINT L1
1290 INPUT "                 ";I,P,U
1300 WRITE #1,I,P,U
1310 U=U*ATN(1)/45
1320 L=2*I-1
1330 A(L,N+1)=-P*COS(U)+A(L,N+1)
1340 A(L+1,N+1)=-P*SIN(U)+A(L+1,N+1)
1350 NEXT L1
1360 GOSUB 4030
1370 REM PRINT "Joint No.          X              Y"
1380 REM FOR I=1 TO N1
1390 REM PRINT TAB(8) I,TAB(17) X(I),TAB(27) Y(I)
1400 REM NEXT I
1410 PRINT "*** OUTPUT DATA ***"
1420 PRINT #1,"*** OUTPUT DATA ***"
1430 C$ = "THE FOLLOWING DATA ARE THE ANSWERS: ";PRINT C$ ;PRINT #1,C$
1440 PRINT "INTERNAL FORCES IN EACH BAR:"
1450 PRINT #1,"INTERNAL FORCES IN EACH BAR:"
1460 FOR I=1 TO N2
1470 PRINT TAB(5) "S ";I;"=";A(I,N+1)
1480 REM PRINT TAB(6) I,TAB(25) "S";I;"=";A(I,N+1)
```

```
1490 PRINT #1,"      S";RIGHT$(STR$(I),2);"=";A(I,N+1)
1500 NEXT I
1510 PRINT "SUPPORTING FORCES:"
1520 PRINT #1,"SUPPORTING FORCES:"
1530 FOR I=1 TO N3
1540 PRINT "    R";RIGHT$(STR$(I),2);"=";A(N2+I,N+1)
1550 PRINT #1,"     R";RIGHT$(STR$(I),2);"=";A(N2+I,N+1)
1560 NEXT
1570 GOTO 1600
1580 PRINT "2(TOTAL JOINT NO.)<>(TOTAL SUPPORTING FORCES No.)+(TOTAL BAR NO.)"
1590 PRINT "CHECK YOUR INPUT DATA PLEASE."
1600 RETURN
1610 REM -----------------------------------------------------------------------SPACIAL TRUSSES
1620 REM
1630 PRINT "THIS IS THE SUBROUTINE OF SPACIAL TRUSSES FROM NOW ON."
1640 GOSUB 670
1650 N=3*N1
1660 N4=N2+N3
1670 IF N<> N4 THEN 2400
1680 PRINT "ENTER EACH JOINT COORDINATE:   X,Y,Z          "
1690 FOR I=1 TO N1
1700 PRINT "JOINT";I
1710 INPUT "                  ";X(I),Y(I),Z(I)
1720 WRITE #1,X(I),Y(I),Z(I)
1730 NEXT I
1740 PRINT "ENTER TWO JOINT NUMBERS IN EACH BAR:"
1750 PRINT "BAR No.       JOINT No.    (A,B)        "
1760 FOR I=1 TO N2
1770 PRINT "     ";I
1780 INPUT "                  ";J,K
1790 WRITE #1,J,K
1800 X1=X(K)-X(J)
1810 Y1=Y(K)-Y(J)
1820 Z1=Z(K)-Z(J)
1830 D=SQR(X1*X1+Y1*Y1+Z1*Z1)
1840 C1=X1/D
1850 C2=Y1/D
1860 C3=Z1/D
1870 L=3*J-2
1880 A(L,I)=C1
1890 A(L+1,I)=C2
1900 A(L+2,I)=C3
1910 L=3*K-2
1920 A(L,I)=-C1
1930 A(L+1,I)=-C2
1940 A(L+2,I)=-C3
1950 NEXT I
1960 PRINT "ENTER THE ACTING JOINT NO. AND THE RESPECTIVE ANGLE"
1970 PRINT "   WITH X,Y,Z AXIS OF EACH SUPPORTING FORCE."
1980 PRINT "SUPPORTING FORCE NO.,JOINT NO.,ANGLE(X),ANGLE(Y),ANGLE(Z)  "
1990 E=ATN(1)/45
```

```
2000 FOR I=1 TO N3
2010 PRINT "                  ";I
2020 INPUT "                          ";J,U,V,W
2030 WRITE #1,J,U,V,W
2040 L=3*J-2
2050 M=N2+I
2060 A(L,M)=COS(U*E)
2070 A(L+1,M)=COS(V*E)
2080 A(L+2,M)=COS(W*E)
2090 NEXT I
2100 INPUT "*** HOW MANY KNOWN-FORCES ARE THERE",M1
2110 WRITE #1,M1
2120 IF M1<=0 THEN 2230
2130 PRINT "    JOINT NO.,MAGNITUDE,ANGLE(X),ANGLE(Y),ANGLE(Z)  "
2140 FOR M=1 TO M1
2150 PRINT M
2160 INPUT "               ";I,P,U,V,W
2170 WRITE #1,I,P,U,V,W
2180 L=3*I-2
2190 A(L,N+1)=-P*COS(U*E)+A(L,N+1)
2200 A(L+1,N+1)=-P*COS(V*E)+A(L+1,N+1)
2210 A(L+2,N+1)=-P*COS(W*E)+A(L+2,N+1)
2220 NEXT M
2230 GOSUB 4030
2240 PRINT C$
2250 PRINT #1,"*** OUTPUT DATA ***"
2260 PRINT    "*** OUTPUT DATA ***"
2270 PRINT #1,"INTERNAL FORCES OF EACH BAR:"
2280 PRINT    "INTERNAL FORCES OF EACH BAR:"
2290 FOR I=1 TO N2
2300 PRINT    "   S";RIGHT$(STR$(I),2);"=";A(I,N+1)
2310 PRINT #1,"   S";RIGHT$(STR$(I),2);"=";A(I,N+1)
2320 NEXT I
2330 PRINT    "SUPPORTING FORCES:"
2340 PRINT #1,"SUPPORTING FORCES:"
2350 FOR I=1 TO N3
2360 PRINT    "   R";RIGHT$(STR$(I),2);"=";A(N2+I,N+1)
2370 PRINT #1,"   R";RIGHT$(STR$(I),2);"=";A(N2+I,N+1)
2380 NEXT I
2390 GOTO 2430
2400 PRINT "3(TOTAL JOINT NO.)<>(TOTAL SUPPORTING FORCES NO.)+"
2410 PRINT "           (TOTAL BAR NO)"
2420 PRINT "CHECK YOUR DATA PLEASE."
2430 RETURN
2440 REM-------------------------------------------------------------------------------PLANAR FORCE SYSTEMS.
2450 REM
2460 PRINT "THIS IS THE SUBROUTINE OF PLANAR FORCE SYSTEMS FROM NOW ON."
2470 GOSUB 740
2480 N=3*N2
2490 IF N4 >N THEN 3190
2500 FOR I=1 TO N2
```

```
2510 PRINT
2520 PRINT "RIGID BODY";I;"#"
2530 PRINT
2540 INPUT " * * * HOW MANY UNKNOWN-FORCES ARE THERE";M3
2550 WRITE #1,M3
2560 IF M3<=0 THEN 2700
2570 PRINT "UNKNOWN-FORCE NO. ,ANGLE(X),    X,      Y"
2580 L=3*I-2
2590 FOR J1=1 TO M3
2600 INPUT "                              ",J,U,X,Y
2610 WRITE #1,J,U,X,Y
2620 U=U*ATN(1)/45
2630 C=COS(U)
2640 S=SIN(U)
2650 A(L,J)=C
2660 A(L+1,J)=S
2670 A(L+2,J)=X*S-Y*C
2680 NEXT J1
2690 PRINT "ENTER UNKNOWN-MOMENTS;"
2700 INPUT " * * * HOW MANY UNKNOWN-MOMENTS ARE THERE";M4
2710 WRITE #1,M4
2720 IF M4<=0 THEN 2800
2730 FOR N1=1+M3 TO M4+M3
2740 PRINT "ENTER UNKNOWN-MOMENT NO. AND ROTARY DIRECTION CODE OF"
2750 PRINT "UNKNOWN-MOMENT (+1,FOR COUNTERCLOCKWISE;-1,FOR CLOCKWISE);";
2760 INPUT J,X
2770 WRITE #1,J,X
2780 A(L+2,J)=SGN(X)+A(L+2,J)
2790 NEXT N1
2800 PRINT "ENTER KNOWN-FORCES;"
2810 INPUT " * * * HOW MANY KNOWN-FORCES TO BE ENTERED?",M1
2820 WRITE #1,M1
2830 IF M1<=0 THEN 2960
2840 PRINT "   MAGNITUDE,ANGLE(X),COORDINATE (X,Y)"
2850 FOR M=1 TO M1
2860 PRINT "No. ";M
2870 INPUT "          ";P,U,X,Y
2880 WRITE #1,P,U,X,Y
2890 U=U*ATN(1)/45
2900 C=-P*COS(U)
2910 S=-P*SIN(U)
2920 A(L,N+1)=C+A(L,N+1)
2930 A(L+1,N+1)=S+A(L+1,N+1)
2940 A(L+2,N+1)=X*S-Y*C+A(L+2,N+1)
2950 NEXT M
2960 PRINT "ENTER NUMBER AND ALGEBRAIC VALUE OF KNOWN-MOMENTS;"
2970 INPUT " * * * HOW MANY KNOWN-MOMENTS TO BE ENTERED?",M2
2980 WRITE #1,M2
2990 IF M2<=0 GOTO 3060
3000 PRINT "ALGEBRAIC VALUE OF KNOWN-MOMENTS;";
3010 FOR M=1 TO M2
```

```
3020 INPUT P
3030 WRITE #1,P
3040 A(L+2,N+1)=-P+A(L+2,N+1)
3050 NEXT M
3060 NEXT I
3070 GOSUB 4030
3080 F$="OF THE UNKNOWN-FORCES AND UNKNOWN-MOMENTS."
3090 PRINT "THE FOLLOWING DATA ARE THE ANSWERS OF UNKNOWN FORCES AND MOMENTS."
3100 PRINT " * * * OUTPUT DATA * * *"
3110 PRINT #1," * * * OUTPUT DATA * * *"
3120 PRINT "CONSTRAINED FORCES:"
3130 PRINT #1,"CONSTRAINED FORCES:"
3140 FOR I=1 TO N4
3150 PRINT "      R";RIGHT$(STR$(I),2);"=";A(I,N+1)
3160 PRINT #1,"      R";RIGHT$(STR$(I),2);"=";A(I,N+1)
3170 NEXT I
3180 GOTO 3200
3190 PRINT "YOU ENTERED THE WRONG DATA,CHECK YOUR INPUT DATA CAREFULLY."
3200 RETURN
3210 REM ------------------------------------------------------------------------------SPACIAL FORCE SYSTEM.
3220 REM
3230 PRINT "THIS IS THE SUBROUTINE OF SPACIAL FORCE SYSTEMS FROM NOW ON."
3240 GOSUB 740
3250 N=6*N2
3260 IF N4>N THEN 4010
3270 E=ATN(1)/45
3280 FOR I=1 TO N2
3290 PRINT
3300 PRINT "RIGID BODY" I;"#"
3310 PRINT
3320 PRINT "ENTER UNKNOWN-FORCE NO.,ANGLE(X),ANGLE(Y),ANGLE(Z),COORDINATE X,Y,Z:"
3330 L=6*I-5
3340 INPUT " * * * HOW MANY UNKNOWN-FORCE ARE THERE";M3
3350 WRITE #1,M3
3360 IF M3<=0 GOTO 3510
3370 D$=" No. OF UNKNOWN-FORCES,ANGLE(X),ANGLE(Y),ANGLE(Z),X,Y,Z":PRINT D$
3380 FOR J1=1 TO M3
3390 INPUT "      ";J,U,V,W,X,Y,Z
3400 WRITE #1,J,U,V,W,X,Y,Z
3410 C1=COS(U*E)
3420 C2=COS(V*E)
3430 C3=COS(W*E)
3440 A(L,J)=C1
3450 A(L+1,J)=C2
3460 A(L+2,J)=C3
3470 A(L+3,J)=Y*C3-Z*C2
3480 A(L+4,J)=Z*C1-X*C3
3490 A(L+5,J)=X*C2-Y*C1
3500 NEXT J1
3510 INPUT " * * * HOW MANY UNKNOWN-MOMENTS ARE THERE";M4
3520 WRITE #1,M4
```

```
3530 IF M4<=0 GOTO 3620
3540 E$ ="NO. OF UNKNOWN-MOMENTS,ANGLE(X),ANGLE(Y),ANGLE(Z)";PRINT E$
3550 FOR J1=M3+1 TO M4+M3
3560 INPUT "            ";J,U,V,W
3570 WRITE #1,J,U,V,W
3580 A(L+3,J)=COS(U*E)+A(L+3,J)
3590 A(L+4,J)=COS(V*E)+A(L+4,J)
3600 A(L+5,J)=COS(W*E)+A(L+5,J)
3610 NEXT J1
3620 INPUT " * * * HOW MANY KNOWN-FORCES ARE THERE";M5
3630 WRITE #1,M5
3640 IF M5<=0 GOTO 3790
3650 PRINT "MAGNITUDE OF KNOWN-FORCE,ANGLE(X),ANGLE(Y),ANGLE(Z),X,Y,Z"
3660 FOR M=1 TO M5
3670 INPUT P,U,V,W,X,Y,Z
3680 WRITE #1,P,U,V,W,X,Y,Z
3690 C1=-P*COS(U*E)
3700 C2=-P*COS(V*E)
3710 C3=-P*COS(W*E)
3720 A(L,N+1)=C1+A(L,N+1)
3730 A(L+1,N+1)=C2+A(L+1,N+1)
3740 A(L+2,N+1)=C3+A(L+2,N+1)
3750 A(L+3,N+1)=Y*C3-Z*C2+A(L+3,N+1)
3760 A(L+4,N+1)=Z*C1-X*C3+A(L+4,N+1)
3770 A(L+5,N+1)=X*C2-Y*C1+A(L+5,N+1)
3780 NEXT M
3790 INPUT " * * * HOW MANY KNOWN-MOMENTS ARE THERE";M6
3800 WRITE #1,M6
3810 IF M6<=0 GOTO 3910
3820 PRINT "MAGNITUDE";E$
3830 FOR M=1 TO M6
3840 INPUT P,U,V,W
3850 WRITE #1,P,U,V,W
3860 A(L+3,N+1)=-P*COS(U*E)+A(L+3,N+1)
3870 A(L+4,N+1)=-P*COS(V*E)+A(L+4,N+1)
3880 A(L+5,N+1)=-P*COS(W*E)+A(L+5,N+1)
3890 NEXT M
3900 NEXT I
3910 GOSUB 4030
3920 PRINT " * * * OUTPUT DATA * * *"
3930 PRINT #1," * * * OUTPUT DATA * * *"
3940 PRINT C$
3950 PRINT #1,C$
3960 FOR I=1 TO N4
3970 PRINT "     R";RIGHT$ (STR$ (I),2);"=";A(I,N+1)
3980 PRINT #1,"     R";RIGHT$ (STR$ (I),2);"=";A(I,N+1)
3990 NEXT I
4000 GOTO 4020
4010 PRINT "YOU ENTERED THE WRONG DATA,CHECK YOUR INPUT DATA CAREFULLY."
4020 RETURN
4030 FOR I=1 TO N4
```

```
4040 G(I)=I
4050 NEXT I
4060 E2=0
4070 FOR I=1 TO N
4080 E2=E2+ABS(A(I,N+1))
4090 NEXT I
4100 E2=E2*.000001
4110 FOR K=1 TO N4
4120 C=0
4130 FOR I=K TO N
4140 FOR J=K TO N4
4150 IF ABS(A(I,J))<=ABS(C) THEN 4190
4160 C=A(I,J)
4170 I1=I
4180 J1=J
4190 NEXT J
4200 NEXT I
4210 IF ABS(C)<=EP THEN 4630
4220 IF J1=K THEN 4310
4230 FOR I=1 TO N
4240 T=A(I,J1)
4250 A(I,J1)=A(I,K)
4260 A(I,K)=T
4270 NEXT I
4280 J=G(K)
4290 G(K)=G(J1)
4300 G(J1)=J
4310 IF I1=K THEN 4370
4320 FOR J=1 TO N+1
4330 T=A(I1,J)
4340 A(I1,J)=A(K,J)
4350 A(K,J)=T
4360 NEXT J
4370 C=1/C
4380 FOR J=K+1 TO N+1
4390 A(K,J)=A(K,J)*C
4400 NEXT J
4410 FOR I=K+1 TO N
4420 FOR J=K+1 TO N+1
4430 A(I,J)=A(I,J)-A(I,K)*A(K,J)
4440 NEXT J
4450 NEXT I
4460 NEXT K
4470 IF N=N4 THEN 4510
4480 FOR I=N4+1 TO N
4490 IF ABS(A(I,N+1))>E2 THEN 4630
4500 NEXT I
4510 FOR I=N4 TO 1 STEP-1
4520 FOR J=I+1 TO N4
4530 A(I,N+1)=A(I,N+1)-A(I,J)*A(J,N+1)
4540 NEXT J
```

```
4550 NEXT I
4560 FOR K=1 TO N4
4570 A(G(K),1)=A(K,N+1)
4580 NEXT K
4590 FOR K=1 TO N4
4600 A(K,N+1)=A(K,1)
4610 NEXT K
4620 GOTO 4640
4630 PRINT "EQUATIONS HAVE NO SOLUTION. "
4640 RETURN
4650 REM -----------------------------------------------------------------------SLIDFRIC. BAS
4660 PRINT "THE SUBROUTINE OF SLIDING FRICTION FROM NOW ON."
4670 REM EQUILIBRIUM PROBLEMS CONSIDERING SLIDING FRICTION.
4680 REM FINDING;UNKNOWN NORMAL REACTIONS & FRICTIONS.
4690 PRINT "* * * * * * * * * * * * * * * * *INPUT* * * * * * * * * * * * * * * * * * *"
4700 PRINT
4710 INPUT "NUMBER OF MOMENT CENTRE NC=";NC
4720 DIM AA(NC,NC+1),MO(NC),LF(NC,NC),LR(NC,NC),GL(NC,NC),U(5),R(9)
4730 PI=3.14159
4740 RA=PI/180
4750 M=0
4760 FOR I=1 TO NC
4770   PRINT "ABOUT MOMENT CENTRE";I
4780   PRINT "    MOMENT OF GIVEN COUPLE M=";:INPUT M
4790   PRINT "    NUMBER OF GIVEN FORCE N=";:INPUT N
4800   IF N=0 THEN 4880
4810   FOR J=1 TO N
4820     PRINT "    MAGNITUDE OF GIVEN FORCE F";J;"=";
4830     INPUT F
4840     PRINT "    FORCE ARM OF GIVEN FORCE ";"L";J;"=";
4850     INPUT L
4860     M=M+F*L
4870   NEXT J
4880   MO(I)=M
4890   AA(I,NC+1)=-MO(I)
4900 NEXT I
4910 INPUT "COEFFICIENT OF STATIC SLIDING FRICTION CF=";CF
4920 INPUT "NUMBER OF SLIDING FRICTION NF=";NF
4930 FOR I=1 TO NC
4940   FOR J=1 TO NC
4950   PRINT "ARM OF No.";J;"UNKNOWN REACTION ABOUT No.";I
4960   PRINT "    MOMENT CENTRE LR(";I;",";J;")=";
4970   INPUT LR(I,J)
4980   NEXT J
4990 NEXT I
5000 FOR I=1 TO NC
5010   FOR J=1 TO NC
5020   PRINT "ARM OF No.";J;"UNKNOWN FRICTION ABOUT No.";I
5030   PRINT "MOMENT CENTRE LF(";I;",";J;")=";
5040   INPUT LF(I,J)
5050   NEXT J
```

```
5060 NEXT I
5070 FOR I=1 TO NC
5080   FOR J=1 TO NC
5090     GL(I,J)=LR(I,J)+CF*LF(I,J)
5100   NEXT J
5110 NEXT I
5120 FOR I=1 TO NC
5130   FOR J=1 TO NC
5140     AA(I,J)=GL(I,J)
5150   NEXT J
5160 NEXT I
5170 IF NC=2 THEN GOSUB 5410
5180 IF NC=3 THEN GOSUB 5610
5190 PRINT
5200 PRINT "*******************OUTPUT*********************"
5210 PRINT #1,"*******************OUTPUT*********************"
5220 PRINT
5230 PRINT #1,
5240 PRINT "SOLUTION OF EQUATIONS:-"
5250 PRINT #1,"SOLUTION OF EQUATIONS:-"
5260 PRINT TAB(3)"MOMENT CENTRE No.";
5270 PRINT #1,TAB(3)"MOMENT CENTRE No.";
5280 PRINT TAB(25)"NORMAL REACTION";TAB(50)"FRICTION"
5290 PRINT #1,TAB(25)"NORMAL REACTION";TAB(50)"FRICTION"
5300 FOR I=1 TO NC
5310   PRINT TAB(5);I;TAB(27);INT(R(I)*1000+.5)/1000;
5320   PRINT #1,TAB(5);I;TAB(27);INT(R(I)*1000+.5)/1000;
5330   IF I=NF+1 THEN 5360
5340   PRINT TAB(52);INT(R(I)*CF*1000+.5)/1000
5350   PRINT #1,TAB(52);INT(R(I)*CF*1000+.5)/1000
5360 NEXT I
5370 PRINT
5380 END
5390 REM SUBROUTINE:SOLVING LINEAR EQUATION SYSTEMS
5400 REM USING CRAMER'S RULE.
5410 GOSUB 5570
5420 D=T
5430 IF ABS(D)<--7665.988 THEN PRINT
5440 FOR Q=1 TO 2
5450   U(1)=AA(1,Q)
5460   U(2)=AA(2,Q)
5470   AA(1,Q)=AA(1,3)
5480   AA(2,Q)=AA(2,3)
5490   GOSUB 5570
5500   R(Q)=T/D
5510   IF Q=2 THEN RETURN
5520   AA(1,Q)=U(1)
5530   AA(2,Q)=U(2)
5540 NEXT Q
5550 RETURN
5560 REM SUBROUTINE
```